

Chapter 35

The LATTICE Procedure

Chapter Table of Contents

OVERVIEW	1747
GETTING STARTED	1747
SYNTAX	1750
PROC LATTICE Statement	1750
VAR Statement	1750
BY Statement	1751
DETAILS	1751
Input Data Set	1751
Missing Values	1752
Displayed Output	1752
ODS Table Names	1753
EXAMPLE	1754
Example 35.1 Analysis of Variance through PROC LATTICE	1754
REFERENCES	1757

Chapter 35

The LATTICE Procedure

Overview

The LATTICE procedure computes the analysis of variance and analysis of simple covariance for data from an experiment with a lattice design. PROC LATTICE analyzes balanced square lattices, partially balanced square lattices, and some rectangular lattices.

In balanced square lattices, the number of treatments is equal to the square of the number of units per block. Incomplete blocks are grouped to form mutually orthogonal replications. The number of replicates in the basic plan is always 1 plus the number of units per block.

The partially balanced square lattices are similar to balanced lattices, although the number of replicates may vary. Partially balanced designs are constructed of the replicates in the basic plan, but not all replicates are included the same number of times, and some may not be included at all.

In rectangular lattices, there are k units per block and $k(k+1)$ treatments. As in square lattices, blocks are grouped to form mutually orthogonal replicates in the basic plan. PROC LATTICE can analyze simple rectangular lattices (two orthogonal replications) and triple rectangular lattices (three orthogonal replications). The experiment can include several repetitions of the basic plan.

The LATTICE procedure determines from the data set which type of design has been used. It also checks to see if the design is valid and displays an appropriate message if it is not.

Getting Started

An example of a balanced square design is an experiment to investigate the effects of nine diets on the growth rate of pigs.

In some breeds of pigs, past experience has shown that a large part of the total variation in growth rates between animals can be attributed to the litter. Therefore, this experiment is planned so that litter differences do not contribute to the intrablock error.

First, the pigs are separated into sets of three litter-mates. Each block is assigned two sets of the three litter-mates. In a given block, one pig from each set receives a diet. Therefore, the experimental unit is a pair of pigs feeding in a particular pen on one of the nine diets. The response variable, growth rate, is the sum of the growth rates for the two pigs in a particular pen. To get the adjusted diet mean per pig, the adjusted treatment mean for the pen must be divided by 2.

The special numeric SAS variables named **Group**, **Block**, **Treatmnt**, and **Rep** must be used to define the design. In this example, the **Treatmnt** variable ranges from 1 to 9 and indicates the particular diet. The **Block** variable is 1, 2, or 3 and indicates the pen containing the two pigs. The **Group** variable ranges from 1 to 4 and specifies which replication within the basic plan includes the experimental unit. In this example, you would not use the **Rep** variable since the entire basic plan is not replicated.

You can use the following **DATA** step and **PROC LATTICE** statement to analyze this experiment. The response variable is **Weight**.

```

title 'Examining the Growth Rate of Pigs';
data Pigs;
  input Group Block Treatmnt Weight @@;
  datalines;
1 1 1 2.20  1 1 2 1.84  1 1 3 2.18
1 2 4 2.05  1 2 5 0.85  1 2 6 1.86
1 3 7 0.73  1 3 8 1.60  1 3 9 1.76
2 1 1 1.19  2 1 4 1.20  2 1 7 1.15
2 2 2 2.26  2 2 5 1.07  2 2 8 1.45
2 3 3 2.12  2 3 6 2.03  2 3 9 1.63
3 1 1 1.81  3 1 5 1.16  3 1 9 1.11
3 2 2 1.76  3 2 6 2.16  3 2 7 1.80
3 3 3 1.71  3 3 4 1.57  3 3 8 1.13
4 1 1 1.77  4 1 6 1.57  4 1 8 1.43
4 2 2 1.50  4 2 4 1.60  4 2 9 1.42
4 3 3 2.04  4 3 5 0.93  4 3 7 1.78
;

proc lattice data=Pigs;
  var Weight;
run;

```

The SAS code produces the output shown in Figure 35.1.

Examining the Growth Rate of Pigs			
The Lattice Procedure			
Analysis of Variance for Weight			
Source	DF	Sum of Squares	Mean Square
Replications	3	0.07739	0.02580
Blocks within Replications (Adj.)	8	1.4206	0.1776
Component B	8	1.4206	0.1776
Treatments (Unadj.)	8	3.2261	0.4033
Intra Block Error	16	1.2368	0.07730
Randomized Complete Block Error	24	2.6574	0.1107
Total	35	5.9609	0.1703

Additional Statistics for Weight	
Description	Value
Variance of Means in Same Block	0.04593
LSD at .01 Level	0.6259
LSD at .05 Level	0.4543
Efficiency Relative to RCBD	120.55

Examining the Growth Rate of Pigs	
The Lattice Procedure	
Adjusted Treatment Means for Weight	
Treatment	Mean
1	1.8035
2	1.7544
3	1.9643
4	1.7267
5	0.9393
6	1.8448
7	1.3870
8	1.4347
9	1.5004

Figure 35.1. Output from Example LATTICE Procedure

Diet 3 yields the highest mean growth rate at 1.9643 pounds for the two pigs (0.9822 per pig), while diet 5 has the lowest rate at 0.9393 (0.4696 per pig). The efficiency of the experiment relative to a randomized complete block design is 120.55 percent, so using the lattice design increased precision, producing more accurate estimates of the treatment effects. The different elements of the LATTICE procedure's output are discussed in the "Displayed Output" section on page 1752.

Syntax

The following statements are available in PROC LATTICE.

```
PROC LATTICE < options >;  
  BY variables ;  
  VAR variables ;
```

Three specific numeric SAS variables, **Group**, **Block**, and **Treatmnt**, *must* be present in the data set to which PROC LATTICE is applied. A fourth numeric variable named **Rep** must be present when the design involves repetition of the entire basic plan. (See the “Input Data Set” section on page 1751 for more information.)

Every numeric variable other than **Group**, **Block**, **Treatmnt**, or **Rep** in the input SAS data set may be considered a response variable. A **VAR** statement tells PROC LATTICE that only the variables listed in the **VAR** statement are to be considered response variables. If the **VAR** statement is omitted, then all numeric variables, excluding **Group**, **Block**, **Treatmnt**, and **Rep**, are considered response variables. PROC LATTICE performs an analysis for each response variable.

PROC LATTICE Statement

```
PROC LATTICE < options >;
```

You can specify the following options in the PROC LATTICE statement.

DATA=SAS-data-set

names the SAS data set to be used by PROC LATTICE. If you omit the **DATA=** option, the most recently created SAS data set is used.

COVARIANCE

COV

calculates sums of products for every possible pair of response variables. A sum of products is given for each source of variation in the analysis of variance table. For each pair of response variables, the one appearing later in the data set (or in the **VAR** statement) is the covariable.

VAR Statement

```
VAR variables ;
```

The **VAR** statement specifies the response variables. If you do not include a **VAR** statement, all numeric variables in the data set are considered response variables (except **Group**, **Block**, **Treatmnt**, and **Rep**).

BY Statement

BY variables ;

You can specify a BY statement with PROC LATTICE to obtain separate analyses on observations in groups defined by the BY variables. When a BY statement appears, the procedure expects the input data set to be sorted in order of the BY variables. The *variables* are one or more variables in the input data set.

If your input data set is not sorted in ascending order, use one of the following alternatives:

- Sort the data using the SORT procedure with a similar BY statement.
- Specify the BY statement option NOTSORTED or DESCENDING in the BY statement for the ANOVA procedure. The NOTSORTED option does not mean that the data are unsorted but rather that the data are arranged in groups (according to values of the BY variables) and that these groups are not necessarily in alphabetical or increasing numeric order.
- Create an index on the BY variables using the DATASETS procedure (in base SAS software).

For more information on the BY statement, refer to the discussion in *SAS Language Reference: Concepts*. For more information on the DATASETS procedure, refer to the discussion in the *SAS Procedures Guide*.

Details

Input Data Set

Four numeric SAS variables, **Group**, **Block**, **Treatmnt**, and **Rep**, are used in the input data set to define the lattice design. The **Group**, **Block**, and **Treatmnt** variables are required in the data set to which PROC LATTICE is applied. The **Rep** variable must be present when the design involves repetition of the entire basic plan.

Group	specifies which orthogonal replication in the basic plan includes the experimental unit. Values of Group must be 1, 2, . . . , n , where n is the number of replicates in the basic plan.
Block	specifies the block in which the experimental unit is present. Values of Block must be 1, 2, . . . , m , where m is the number of blocks in a replication.
Treatmnt	specifies which treatment was applied to the experimental unit. Values of Treatmnt must be 1, 2, . . . , i , where i is the number of treatments in a replication.

Rep specifies which repetition of the basic plan includes the experimental unit. Values of **Rep** must be 1, 2, ..., p , where p is the number of replications of the entire basic plan. Thus, the experiment has a total of np replicates.

Missing Values

If a value of **Group**, **Block**, **Treatmnt**, or **Rep** is missing, the analysis is not performed and an appropriate error message is displayed.

If a value of a response variable is missing, this entire variable is dropped from the analysis. If other response variables exist that do not have missing values, they are analyzed.

Displayed Output

For each response variable, PROC LATTICE displays

- an “Analysis of Variance” table and related statistics, including the following as separate sources of variations:
 - Replications
 - Blocks within Replications (adjusted for treatments)
 - Treatments (unadjusted)
 - Intra Block Error
 - Randomized Complete Block Error

The Blocks within Replications sum of squares is further broken down into “Component A” and “Component B.” If there is no repetition of the basic plan, the Component B sum of squares is the same as the Blocks within Replications sum of squares. If there is repetition of the basic plan, the Component A sum of squares reflects the variation among blocks that contain the same treatments.

The source of variation called Randomized Complete Block Error is the sum of the Blocks within Replications sum of squares and the Intra Block Error sum of squares. It is the appropriate error term if the experimental design is a randomized complete block design, with the replications filling the roles of complete blocks.

- two values for the Variance of Means. For some lattice designs, these are only approximations. The first value is applicable when the two treatments appear in the same block; the other (when it appears) applies when the two treatments never appear in the same block (a possibility in partially balanced and rectangular designs).
- an Average of Variance. Except with small designs, it is sufficient to use this average variance of means for tests between treatments (whether the two treatments appear in the same block or not); refer to Cochran and Cox (1957).
- the Least Significant Differences (LSDs) at the .01 and .05 levels of significance, based on the Average of Variance

- Efficiency Relative to RCBD, the efficiency of the lattice design relative to a randomized complete block design. The efficiency is the ratio of the randomized complete block mean squared error to the effective error variance; refer to Cochran and Cox (1957).
- the Adjusted Treatment Means. These are adjusted for blocks if the relative precision is greater than 105 percent.

When you specify the COVARIANCE option, PROC LATTICE produces Sums of Products and the Mean Product for each source of variation in the analysis of variance table.

ODS Table Names

PROC LATTICE assigns a name to each table it creates. You can use these names to reference the table when using the Output Delivery System (ODS) to select tables and create output data sets. These names are listed in the following table. For more information on ODS, see Chapter 15, “Using the Output Delivery System.”

Table 35.1. ODS Tables Produced in PROC LATTICE

ODS Table Name	Description	PROC LATTICE Option
ANOVA	Analysis of variance	default
AdjTreatmentMeans	Adjusted treatment means	default
Statistics	Additional statistics	default

Example

Example 35.1. Analysis of Variance through PROC LATTICE

In the following example, from Cochran and Cox (1957, p. 406), the data are yields (Yield) in bushels per acre of 25 varieties (Treatment) of soybeans. The data are collected in two replications (Group) of 25 varieties in five blocks (Block) containing five varieties each. This is an example of a partially balanced square lattice design.

```

data Soy;
  do Group = 1 to 2;
    do Block = 1 to 5;
      do Plot = 1 to 5;
        input Treatment Yield @@;
        output;
      end;
    end;
  end;
  drop Plot;
  datalines;
1   6   2   7   3   5   4   8   5   6
6   16  7  12  8  12  9  13  10  8
11  17  12  7  13  7  14  9  15  14
16 18  17 16 18 13 19 13 20 14
21 14 22 15 23 11 24 14 25 14
1   24  6  13 11 24 16 11 21  8
2   21  7  11 12 14 17 11 22 23
3   16  8   4  13 12 18 12 23 12
4   17  9  10 14 30 19  9  24 23
5   15 10 15 15 22 20 16 25 19
;

proc print data=Soy;
  id Treatment;
run;

proc lattice data=Soy;
run;

```

The results from these statements are shown in Output 35.1.1 and Output 35.1.2.

Output 35.1.1. Displayed Output from PROC PRINT

Treatmnt	Group	Block	Yield
1	1	1	6
2	1	1	7
3	1	1	5
4	1	1	8
5	1	1	6
6	1	2	16
7	1	2	12
8	1	2	12
9	1	2	13
10	1	2	8
11	1	3	17
12	1	3	7
13	1	3	7
14	1	3	9
15	1	3	14
16	1	4	18
17	1	4	16
18	1	4	13
19	1	4	13
20	1	4	14
21	1	5	14
22	1	5	15
23	1	5	11
24	1	5	14
25	1	5	14
1	2	1	24
6	2	1	13
11	2	1	24
16	2	1	11
21	2	1	8
2	2	2	21
7	2	2	11
12	2	2	14
17	2	2	11
22	2	2	23
3	2	3	16
8	2	3	4
13	2	3	12
18	2	3	12
23	2	3	12
4	2	4	17
9	2	4	10
14	2	4	30
19	2	4	9
24	2	4	23
5	2	5	15
10	2	5	15
15	2	5	22
20	2	5	16
25	2	5	19

Output 35.1.2. Displayed Output from PROC LATTICE

The Lattice Procedure			
Analysis of Variance for Yield			
Source	DF	Sum of Squares	Mean Square
Replications	1	212.18	212.18
Blocks within Replications (Adj.)	8	501.84	62.7300
Component B	8	501.84	62.7300
Treatments (Unadj.)	24	559.28	23.3033
Intra Block Error	16	218.48	13.6550
Randomized Complete Block Error	24	720.32	30.0133
Total	49	1491.78	30.4445

Additional Statistics for Yield	
Description	Value
Variance of Means in Same Block	15.7915
Variance of Means in Different Bloc	17.9280
Average of Variance	17.2159
LSD at .01 Level	12.1189
LSD at .05 Level	8.7959
Efficiency Relative to RCBD	174.34

The Lattice Procedure	
Adjusted Treatment Means for Yield	
Treatment	Mean
1	19.0681
2	16.9728
3	14.6463
4	14.7687
5	12.8470
6	13.1701
7	9.0748
8	6.7483
9	8.3707
10	8.4489
11	23.5511
12	12.4558
13	12.6293
14	20.7517
15	19.3299
16	12.6224
17	10.5272
18	10.7007
19	7.3231
20	11.4013
21	11.6259
22	18.5306
23	12.2041
24	17.3265
25	15.4048

The efficiency of the experiment relative to a randomized complete block design is 174.34 percent. Precision is gained using the lattice design via the recovery of intrablock error information, enabling more accurate estimates of the treatment effects. Variety 8 of soybean had the lowest adjusted treatment mean (6.7483 bushels per acre), while variety 11 of soybean had the highest adjusted treatment mean (23.5511 bushels per acre).

References

- Cochran, W.G. and Cox, G.M. (1957), *Experimental Designs*, Second Edition, New York: John Wiley & Sons, Inc.
- Comstock, R.E., Peterson, W.J., and Stewart, H.A. (1948), “An Application of the Balanced Lattice Design in a Feeding Trial with Swine,” *Journal of Animal Science*, 7, 320–331.
- Cornelius, P.L. (1983), “Lattice Designs,” *Encyclopedia of Statistical Sciences*, 4, 510–518.
- Robinson, H.F. and Watson, G.S. (1949), “Analysis of Simple and Triple Rectangular Designs,” *North Carolina Agricultural Experiment Station Technical Bulletin*, 88.

The correct bibliographic citation for this manual is as follows: SAS Institute Inc., *SAS/STAT® User's Guide, Version 8*, Cary, NC: SAS Institute Inc., 1999.

SAS/STAT® User's Guide, Version 8

Copyright © 1999 by SAS Institute Inc., Cary, NC, USA.

ISBN 1-58025-494-2

All rights reserved. Produced in the United States of America. No part of this publication may be reproduced, stored in a retrieval system, or transmitted, in any form or by any means, electronic, mechanical, photocopying, or otherwise, without the prior written permission of the publisher, SAS Institute Inc.

U.S. Government Restricted Rights Notice. Use, duplication, or disclosure of the software and related documentation by the U.S. government is subject to the Agreement with SAS Institute and the restrictions set forth in FAR 52.227-19 Commercial Computer Software-Restricted Rights (June 1987).

SAS Institute Inc., SAS Campus Drive, Cary, North Carolina 27513.

1st printing, October 1999

SAS® and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries.® indicates USA registration.

Other brand and product names are registered trademarks or trademarks of their respective companies.

The Institute is a private company devoted to the support and further development of its software and related services.